マイコン内蔵のＵＡＲＴでの制御信号処理[共通的な内容]

Ｑ１

ＭＰＳＣを用いて、簡単な調歩同期通信を行なってきたが、マイコン内蔵のＵＡＲＴに置き換えを考えている。しかし、内蔵ＵＡＲＴではＣＴＳ信号をサポートしていない。どのようにすればよいのか。

Ａ１

ＵＡＲＴでＲＴＳ信号やＣＴＳ信号のような制御信号を利用するには、出力ポートと外部割込み入力を利用します。

受信では、受信バッファに空きができたらＲＴＳ信号を立ち下げて、送信許可状態であることを示します。逆に受信バッファの空きが少なくなると、送信を止めるように、ＲＴＳ信号を立ち上げます。

送信では、ＣＴＳ信号（送信許可信号）入力は外部割込み（及び入力ポート）として処理します。その立ち下がりエッジ（送信許可）を割り込みで検出してバッファに溜まったデータを送信し、そのハイ・レベルをポートで読むことで新たな送信を停止します。

この関係を下図に示します。


（バッファの制御の概念については、ＦＡＱの「予備知識」に掲載されているバッファ（http://www.necel.com/ja/faq/f_tech.html#0911にリンク）も参照下さい。）

Ｑ２

ＣＴＳなどの制御信号を用いた調歩同期通信を行ないたい。マイコン内蔵のＵＡＲＴを用いた実現方法を知りたい。

Ａ２

ここでは、７８Ｋ０Ｓ／ＫＡ１＋に内蔵されたＵＡＲＴ６を用いた例を考えてみます。ＵＡＲＴ６はシフト・レジスタとバッファ・レジスタの2段バッファ構成になっていますが、複数キャラクタを送受信する場合には別途、送信用と受信用にバッファ・メモリを準備する必要があります。また、通常の処理と送受信処理を並行して行なえるように割り込みを用いた制御を行なうものとします。

ここでは、受信バッファの状態を送信側に知らせるためにＰ４０をＲＴＳ信号（送信要求信号）、Ｐ４１（ＩＮＴＰ３）をＣＴＳ信号（送信許可信号）として使用します。このときの接続例を下図に示します。この例のように、対応した信号同士をクロスで接続します。


バッファ・メモリとしては送受信共に１６バイトの容量を確保し、リング・バッファとして使用します。１６バイトのどれかを示すには４ビットのオフセットを使用します。書き込み用と読み出し用のふたつのオフセットを比較することで、０（バッファが空）から１５（バッファに１５バイトのデータが蓄積されている）までの状態を表すことができます。バッファの確実な制御のために、１６バイト蓄積された状態はオーバフローとして扱い（そのような状態にしない）ます。

受信側でのリング・バッファの構成例を下図に示します。左側に受信処理を示し、右側が受信バッファからの読み出しを示します。受信データのバッファへの書き込みにより、受信バッファに１３バイトのデータが埋まる（読み出し用オフセットから書き込み用オフセットを引いた値（空き）が３キャラクタになる）と、送信要求（ＲＴＳ信号）をオフにします。バッファの空きが３キャラクタ以下になると送信要求をオンにします。ＲＴＳ信号の出力は単純なポート出力として制御します。


送信側では送信要求を送信許可（ＣＴＳ信号）として扱い、オフしたら新たな送信を中止します。ＣＴＳ信号は立ち下がりエッジ検出による割り込みを送信許可（通信再開）のトリガで使用すると共に、次の送信データの書き込み時にポート入力として確認で使用する方法を組み合わせます。

この制御を行なう際の送信側の処理を以下に示します。送信処理では、送信するデータは送信用バッファに一旦蓄積します。その上で送信可能であれば送信処理を起動します。なお、起動処理は単に送信完了割り込み要求フラグをセットするだけです。また受信側からのＣＴＳ信号の立下り割り込みでも起動処理を行ないます。実際のデータ送信は送信完了割り込み（ＩＮＴＳＴ６）処理の中で行ないます。ＩＮＴＳＴ６処理では送信が可能で、送信するデータがあれば次のデータをＵＡＲＴ６の送信バッファに書き込んで送信を行ないます。また、全てのデータの送信が完了すると送信中フラグ（これは起動処理の中でセット）をクリアして処理を終了します。


プログラムは大きく分けて、以下に示す５つの部分から構成されます。

①送信バッファへの送信データの追加（putTxDATA）

②受信バッファから受信データの読み出し（getRxDATA）

③ＣＴＳ立下り割り込み処理（CTSONINT）

④受信完了割り込み処理（RXRDYINT）

⑤送信完了割り込み処理（TXRDYINT）

これ以外に受信エラー割り込み処理とバッファの状態をチェックするサブルーチンがあります。

①送信バッファへの送信データの追加（putTxDATA）

Ａレジスタに送信したいデータをセットしてサブルーチン・コールすることで使用します。バッファ内のデータが１５キャラクタになるとそれ以降はキャリーをセットしてエラーを示します。

１５キャラクタ未満の場合には、書き込むためのオフセット（下位４ビットのみ使用）から書き込むアドレスを計算してそこに送信データを格納します。オフセットを更新（＋１）してバッファ処理は完了し、ＵＡＲＴ６の制御を行ないます。その際に割り込みが入ると余分な送信完了割込みが生じる可能性があるので、以降は排他制御のために割り込み禁止で処理します。ＣＴＳ信号がハイ･レベルであれば、新たな送信は起動しません。そうでないときには、起動中フラグをセットし、送信完了割り込み要求フラグをセットして処理を終了します。


オフセットは下位の４ビットだけが有効なデータですが、更新は８ビットまで行ないます。オフセットから実際のバッファのアドレスを計算するときに下位の４ビットだけを使用するようにしています。

②受信バッファから受信データの読み出し（getRxDATA）

バッファを調べ、バッファが空ならゼロ・フラグをセットしてもどります。データがあれば、残りデータが規定値（BUFFEMP）以下ならＲＴＳ信号をロウ･レベルにして受信可能であることを示します。読み出し用のオフセット（下位ビット）から読み出すデータのアドレスを計算して、バッファからデータを読み出します。その後に、読み出しオフセットを更新して処理を終了します。


③ＣＴＳ立下り割り込み処理（CTSONINT）

ＣＴＳ信号の立下り（受信側のバッファに空きができたことを示す）エッジで発生した割り込みの処理です。ＣＴＳ信号を入力ポートＰ４１から読み、ロウ・レベルかを確認します。再度の確認でロウ・レベルなら、送信許可の処理を行ないます。既に、送信が起動されている（送信中フラグがセットされている）か、送信するデータがなければ、何もせず終了します。送信するデータが残っていれば送信中フラグをセットして、送信完了割り込みを起動します。


④受信完了割り込み処理（RXRDYINT）

受信完了割り込みにより起動され、受信バッファへの書き込み用オフセットから書き込むアドレスを計算し、受信したデータを保存します。オフセットを更新後、バッファ内のデータ数をチェックして、規定値（BUFFFULL）以上ならＲＴＳ信号をハイ・レベルに立ち上げます。


⑤送信完了割り込み処理（TXRDYINT）
実際のデータ送信処理を行なう部分です。データ送信は必ずこの送信完了割り込み処理の中で行われます。処理を開始するとＴＸＢ６が空かを確認（空のはずであるが念のために確認）します。次にＣＴＳ信号を確認してハイ・レベルになっているか、リング・バッファが空になっていれば、新たなデータの送信は行ないません。さらに、ＴＸＳ６が空になっているかを確認し、空なら送信中フラグをクリアします（このときにはＵＡＲＴ６からこれ以上は送信完了の割り込みが発生することはありません。次の起動はputTxDATAでの新しい送信データの書き込みでのプログラムによるセットか、その後のＩＮＴＰ３割り込み処理の中でのプログラムでのセットによるものです）。

送信可能状態で、次に送信すべきデータがリング・バッファに残っていれば、そのデータをＵＡＲＴ６に書き込んで読み出し用のオフセットを更新します。

⑥その他の処理

ＵＡＲＴ６の受信でエラーが発生した場合の処理や送受信のリング・バッファの状態をチェックする処理です。

受信でエラーが発生した場合にはそのときのＡＳＩＳ６の内容をエラー・ステータスとしてセットし、受信したデータは読み飛ばします。

リング・バッファの状態のチェックは書き込みと読み出しの2つのオフセットを比較(書き込み用から読み出し用を引く)、結果の下位４ビットに書き込まれたデータ数を得ています。


リング・バッファ関係のメモリの定義は以下のようになります。初期化の段階でこれらの内容は全て０にクリアしておきます。


また、ベクタ･テーブル関係とプログラムの起動処理関係の例は以下のようになります。使用していない割り込みのベクタにはダミーのアドレスを設定しています。


この後にＵＡＲＴ６を使用するための初期設定が以下の例のように続きます。


⑦使い方

送信する場合にはＡレジスタに送信データをセットしてputTxDATAをコールします。バッファがいっぱいなら、キャリーをみてループさせています。同様に受信ではgetRxDATAをコールします。データがなければゼロ・フラグで確認します。

;


;　送信バッファへのデータの書き込み


;


putTxDATA:


	PUSH	HL


	PUSH	AX


	CALL	!CHKTXDATA	; バッファをチェック


	ADD	A,#0F1H		; バッファ・フルでキャリー


	BC	$TXEXIT		; エラーで終了


	MOV	A,#0FH


	AND	A,TXWTPT	; 下位４ビットを抽出


	ADD	A,#LOW TXBUFF	; オフセットを加算


	MOV	L,A


	MOV	H,#HIGH TXBUFF


	POP	AX		; データを復帰


	PUSH	AX


	MOV	[HL],A		; データをバッファにセット


	INC	TXWTPT		; 書き込みオフセット更新


	DI			; フラグ操作の排他処理のため


;


;　送信が止まっているので、送信可能なら再開する


;


	BT	P4.1,$TXEXIT	; CTSがハイなら再開しない


	SET1	TXFLAG.0	; 送信中フラグをＯＮ


	SET1	STIF6		; 起動のための割り込み要求セット


TXEXIT:


	EI


	POP	AX


	POP	HL


	RET





;


;　受信バッファからデータの読み出し


;


;出力


;データがあるとき


;　Ａレジスタ	：読み出しデータ


;　ZERO		：０


;データが無いとき


;　Ａレジスタ	：０


;　ZERO		：１


;


getRxDATA:


	CALL	!CHKRXDATA	; データ数チェック


	BZ	$GETEXIT	; 空なら分岐


	PUSH	PSW		; フラグをセーブ


	CMP	A,#BUFFEMP	; 残りをチェック


	BNC	$GETNEXT	; 指定値以下なら


	CLR1	P4.0		; ＲＴＳを立ち下げる


GETNEXT:


	PUSH	HL


	MOV	A,#0FH


	AND	A,RXPOINTER	; オフセット計算


	ADD	A,#LOW RXBUFF	; 下位アドレスの計算


	MOV	L,A


	MOV	H,#HIGH	RXBUFF


	MOV	A,[HL]		; データの読み出し


	INC	RXPOINTER	; 読み出しポインタ更新


	POP	HL


	POP	PSW		; フラグを復帰


GETEXIT:


	RET








;


;　ＩＮＴＰ３（立下り検出）割り込み処理


;


;　ＣＴＳが立ち下がり、送信可能になったときの処理


;


CTSONINT:


	BT	P4.1,$CTSONEXIT	; ノイズ対策


	BT	TXFLAG.0,$CTSONEXIT


;


;　送信中ではないときの処理


;


	PUSH	AX


	CALL	!CHKTXDATA	; バッファをチェック


	POP	AX


	BZ	$CTSONEXIT


	SET1	TXFLAG.0	; 送信中フラグをＯＮ


	SET1	STIF6		; 起動のための割り込み要求セット


CTSONEXIT:


	RETI





;


;　ＵＡＲＴ６データ受信処理


;


RXRDYINT:


	PUSH	AX


	PUSH	HL


	MOV	A,RXWTPT	; 受信データ書き込みオフセット


	AND	A,#0FH


	ADD	A,#LOW RXBUFF


	MOV	L,A


	MOV	H,#HIGH RXBUFF	; ポインターをセット


	MOV	A,RXB6		; 受信データ読み出し


	MOV	[HL],A		; データをバッファに保存


	INC	RXWTPT		; オフセットを更新


	POP	HL


	CALL	!CHKRXDATA	; バッファの状態をチェック


	CMP	A,#BUFFFULL


	BC	$RXRDYEXIT	; １３バイト以下なら戻る


	SET1	P4.0		; ＲＴＳを立ち上げる


RXRDYEXIT:


	POP	AX


	RETI





;


;　ＵＡＲＴ６データ送信完了処理


;


TXRDYINT:


	PUSH	AX


	PUSH	HL


	MOV	A,ASIF6


	BT	A.1,$TXRDYEXIT	; 送信中で抜ける


;


; ＴＸＢ６が空ならば、次データ送信のためにＣＴＳをチェックする。


;ＣＴＳがオンなら残りデータの送信を継続し、オフならば現在の


;データの送信完了で送信中ステータスをクリアする。


;





	BT	P4.1,$CTSOFF


	CALL	!CHKTXDATA	; 残りデータをチェック


	BZ	$NODATA


	MOV	A,TXPOINTER	; 出力データ用オフセット読み出し


	AND	A,#0FH


	ADD	A,#LOW TXBUFF


	MOV	L,A


	MOV	H,#HIGH TXBUFF


	MOV	A,[HL]		; データ読み出し


	MOV	TXB6,A		; 送信


	INC	TXPOINTER	; 読み出しオフセット更新


	POP	HL


	POP	AX


	RETI


CTSOFF:


NODATA:


		BT	A.0,$TXRDYEXIT


; 送信動作を完了


	CLR1	TXFLAG.0


TXRDYEXIT:


	POP	HL


	POP	AX


	RETI





;


;　ＵＡＲＴ６受信エラーの処理


;ＡＳＩＳ６レジスタの内容をフラグにコピーするのみ


;


RXERRORINT:


	PUSH	AX


	MOV	A,ASIS6		; ステータスの読み出し


	MOV	RXSTATUS,A


	MOV	A,RXB6		; データのダミー読み出し


	POP	AX


	RETI





;


;　受信データ数確認


;


;出力


;　Ａレジスタ	：バッファされたデータ数


;　ZERO		：空なら1


;


CHKRXDATA:


	MOV	A,RXWTPT	; 書き込みオフセットの読み出し


	SUB	A,RXPOINTER	; 書き込み－読み出し


	AND	A,#0FH		; 有効ビットのみ抽出


	RET





;


;　送信待ちデータ数確認


;


;出力


;　Ａレジスタ	：送信待ちデータ数


;　ZERO		：空なら1


;


CHKTXDATA:


	MOV	A,TXWTPT	; 書き込みオフセットの読み出し


	SUB	A,TXPOINTER	; 書き込み－読み出し


	AND	A,#0FH		; 有効ビットのみ抽出


	RET








;


;　送受信用１６バイトのリング・バッファ領域


;


;


	DSEG	SADDRP


RXBUFF:


	DS	16		; 受信バッファ





TXBUFF:


	DS	16		; 送信バッファ


;


;　リング・バッファのアクセス用オフセットは各バッファで読み出しと


;書き込みを１バイトづつ準備し、その下位４ビットを有効とする


;


RXPOINTER:


	DS	1		; 受信データ読み出しオフセット


RXWTPT:


	DS	1		; 受信データ書き込みオフセット


TXPOINTER:


	DS	1		; 送信データ読み出しオフセット


TXWTPT:


	DS	1		; 送信データ書き込みオフセット





RXSTATUS:


	DS	1		; 受信エラー・ステータス


TXFLAG:


	DS	1		; 送信動作モードフラグ


;　ビット０が１なら送信動作中、０なら送信していない








	DSEG	IHRAM


	DS	20H		; スタック領域確保


STACKTOP:





RSTVECT	CSEG	AT	0


	DW	START		;00:リセットスタート処理のベクタ


INTVECT	CSEG	AT	18H	; ここでは以下の割り込みを定義


	DW	CTSONINT	;18:INTP3


	DW	DUMMY		;1A:INTTM80


	DW	RXERRORINT	;1C:INTSRE6


	DW	RXRDYINT	;1E:INTSR6


	DW	TXRDYINT	;20:INTST6





@@OPTB	CSEG 	AT	0080H


	DB	10011100b


	DB	11111111b





START:


	MOVW	AX,#STACKTOP


	MOVW	SP,AX		; RAMの先頭部分にスタックを設定





;


;　メモリの初期化


;


	MOVW	HL,#RXBUFF


	MOV	A,#0


MEMLOOP:


	MOV	[HL],A


	INC	L


	BNZ	$MEMLOOP





	MOV	CKSR6,#1		; プリスケーラで１／２分周


	MOV	BRGC6,#218		; １／２１８分周する


	MOV	ASIM6,#10010000b	; 受信エラーでＩＮＴＳＲＥ６発生


	MOV	ASICL6,#00000010b


	SET1	TXE6			; 送信許可に設定


	SET1	RXE6			; 必要なら受信許可も設定


	MOV	P4,#00001000b		; 兼用のポートをハイに設定


	CLR1	PM4.3			; TxD6端子を出力モードに設定


;


;　ハンドシェイク関係の初期化


;


	CLR1	PM4.0			; RTS端子を出力モードに設定


	MOV	INTM1,#00		; 立下りエッジに設定


;


;　割り込み関係の初期化


;


	MOV	IF0,#00			; 割り込み要求クリア


	MOV	IF1,#00


	MOV	MK0,#0FFH


	MOV	MK1,#10001011b		; シリアルとＩＮＴＰ３のマスク解除


	EI				; 割り込み許可





;


;　送信の例


;


	MOV	A,#41H


TX15LOOP:


	CALL	!putTxDATA		; データ送信


	BC	$TX15LOOP		; エラーなら繰り返し





;


;　受信の例


;


RXCHARLOOP:


	CALL	!getRxDATA		; データ読み出し


	BZ	$RXCHARLOOP		; データがなければループ








